﻿'----------------------------------------------------------------
'       Class       :   FTPClient.vb
'       Description :   This class will enable a developer to
'                       perform FTP Processing in VB.NET.
'                       The class supports such features as:
'                           - Uploading a file.
'                           - Create/Remove a directory.
'                           - Remove a file.
'                           - and much much more...
'       Date        :   7th February 2002.
'       Conversion  :   The code was converted from C# code.
'       Note        :   If this code works it was converted by 
'                       Vick S.  If it doesn't it was converted
'                       by someone else. :-)
'                       Also note that the code DOES NOT
'                       ACCOMODATE for proxy servers Only
'                       DIRECT connections to FTP Sites.
'
'   Bit Shifting within VB.NET 2002 is not supported via the
'   traditional bit shifting operators (i.e. << - Bitwise left
'   and >> - Bitwise right).
'   Dividing a number by 2^16 in VB.NET 2002 is the same as bit-
'   shifting right 16 positions.
'   Multiplying a number by 2^16 in VB.NET 2002 is the same as
'   bit-shifting left 16 positions.
'
'   Check out the following MSDN Article on bitshifting in 
'   VB.NET 2003:
'       -   Visual Basic .NET 2003 Language Changes
'       -   By Duncan Mackenzie.
'----------------------------------------------------------------
Imports System
Imports System.Net
Imports System.IO
Imports System.Text
Imports System.Net.Sockets
Imports org.codegen.common.ProgressIndicator

' Main FTP Class.
Public Class FTPClient
    Implements IProgressable


#Region "IProgressable implementations"

    Public Event FileUploadStart(ByVal sender As Object, ByVal e As ProgressableEventArgs) _
                Implements IProgressable.processStart

    Public Event FileUploadProgress(ByVal sender As Object, ByVal e As ProgressableEventArgs) _
                    Implements IProgressable.processGoing

    Public Event FileUploadFinished(ByVal sender As Object, ByVal e As ProgressableEventArgs) _
            Implements IProgressable.processFinished

    Public Property wasCancelled() As Boolean Implements IProgressable.wasCancelled

#End Region

#Region "Main Class Variable Declarations"

    Private iBytes As Int32
    Private objClientSocket As Socket

    Private retValue As Int32
    Private loggedIn As Boolean = False
    Private serverMessage, serverReply As String

    ' Set the size of the packet that is used to read and
    '  write data to the FTP Server to the spcified size below.
    Public Const BLOCK_SIZE As Integer = 512
    Private aBuffer(BLOCK_SIZE) As Byte
    Private ASCII As Encoding = Encoding.ASCII

#End Region

#Region "Class Constructors"
    '
    ' Main class constructor.
    Public Sub New()

        Me.RemotePort = 21

    End Sub

    ' Parametized constructor.
    Public Sub New(ByVal sRemoteHost As String, _
                   ByVal sRemotePath As String, _
                   ByVal sRemoteUser As String, _
                   ByVal sRemotePassword As String, _
                   ByVal iRemotePort As Int32)

        Me.RemoteHost = sRemoteHost
        Me.RemotePath = sRemotePath
        Me.RemoteUser = sRemoteUser
        Me.RemotePassword = sRemotePassword
        Me.MessageString = ""
        Me.RemotePort = iRemotePort

    End Sub
#End Region

#Region "Public Properties"

    Public Property RootDir() As String


    Public ReadOnly Property statusMessage(ByVal sFile As String) As String
        Get
            Return " Server:" & Me.RemoteHost & ", File:" & sFile
        End Get
    End Property
    ' Set/Get the name of the FTP Server.
    Public Property RemoteHost() As String

    ' Set/Get the FTP Port Number.
    Public Property RemotePort() As Int32

    ' Set/Get the remote path.
    Public Property RemotePath() As String

    ' Set the remote password.
    Public Property RemotePassword() As String

    ' Set/Get the remote user.
    Public Property RemoteUser() As String

    ' Set the class messagestring.
    Public Property MessageString() As String

#End Region

#Region "Public Subs and Functions"
    '
    ' Return a list of files within a string() array from the
    '  file system.
    Public Function GetFileList(ByVal sMask As String) As String()
        Dim cSocket As Socket
        Dim bytes As Int32
        Dim seperator As Char = ControlChars.Lf
        Dim mess() As String

        serverMessage = ""
        If (Not (Me.loggedIn)) Then
            Login()
        End If

        cSocket = CreateDataSocket()
        SendCommand("NLST " & sMask)

        If (Not (retValue = 150 Or retValue = 125)) Then
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        serverMessage = ""
        Do While (True)
            Array.Clear(aBuffer, 0, aBuffer.Length)
            bytes = cSocket.Receive(aBuffer, aBuffer.Length, 0)
            serverMessage += ASCII.GetString(aBuffer, 0, bytes)

            If (bytes < aBuffer.Length) Then
                Exit Do
            End If
        Loop

        mess = serverMessage.Split(seperator)
        cSocket.Close()
        ReadReply()

        If (retValue <> 226) Then
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        Return mess
    End Function

    Public Function getFileModTime(ByVal sfilename As String) As Date

        Dim modTime As Date
        Dim sModTime As String

        If (Not (Me.loggedIn)) Then
            Login()
        End If

        SendCommand("MDTM " & sfilename)

        If (retValue = 213) Then
            sModTime = serverReply.Substring(4)
            modTime = DateSerial(CInt(sModTime.Substring(0, 4)), CInt(sModTime.Substring(4, 2)), CInt(sModTime.Substring(6, 2)))
            modTime = DateAdd(DateInterval.Hour, CInt(sModTime.Substring(8, 2)), modTime)
            modTime = DateAdd(DateInterval.Minute, CInt(sModTime.Substring(10, 2)), modTime)
            modTime = DateAdd(DateInterval.Second, CInt(sModTime.Substring(12, 2)), modTime)
        Else
            MessageString = serverReply
            'Throw New IOException(m_sReply.Substring(4))
        End If

        Return modTime

    End Function
    '
    ' Get the size of the file on the FTP Server.
    Public Function GetFileSize(ByVal sFileName As String) As Long
        Dim size As Long

        If (Not (Me.loggedIn)) Then
            Login()
        End If

        SendCommand("SIZE " & sFileName)
        size = 0

        If (retValue = 213) Then
            size = Int64.Parse(serverReply.Substring(4))
        Else
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        Return size
    End Function

    '
    ' Log into the FTP Server.
    Public Function Login() As Boolean

        objClientSocket = New Socket(AddressFamily.InterNetwork, SocketType.Stream, ProtocolType.Tcp)
        Dim ep As New IPEndPoint(New System.Net.IPAddress(IPAddress.Parse(Me.RemoteHost).GetAddressBytes()), Me.RemotePort)

        Try
            objClientSocket.Connect(ep)
        Catch ex As Exception
            MessageString = serverReply
            Throw New IOException("Couldn't connect to remote server")
        End Try

        ReadReply()
        If (retValue <> 220) Then
            CloseConnection()
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        SendCommand("USER " & Me.RemoteUser)
        If (Not (retValue = 331 Or retValue = 230)) Then
            Cleanup()
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        If (retValue <> 230) Then
            SendCommand("PASS " & Me.RemotePassword)
            If (Not (retValue = 230 Or retValue = 202)) Then
                Cleanup()
                MessageString = serverReply
                Throw New IOException(serverReply.Substring(4))
            End If
        End If

        Me.loggedIn = True
        ChangeDirectory(Me.RemotePath)

        ' Return the end result.
        Return True
    End Function

    '
    ' If the value of mode is true, set binary mode for 
    '  downloads.
    ' Else, set Ascii mode.
    Public Sub SetBinaryMode(ByVal bMode As Boolean)

        If (bMode) Then
            SendCommand("TYPE I")
        Else
            SendCommand("TYPE A")
        End If

        If (retValue <> 200 And retValue <> 226) Then
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

    End Sub

    '
    ' Download a file to the Assembly's local directory,
    ' keeping the same file name.
    Public Sub DownloadFile(ByVal sFileName As String)
        DownloadFile(sFileName, "", False)
    End Sub

    '
    ' Download a remote file to the Assembly's local 
    '  directory, keeping the same file name, and set 
    '  the resume flag.
    Public Sub DownloadFile(ByVal sFileName As String, _
                            ByVal bResume As Boolean)
        DownloadFile(sFileName, "", bResume)
    End Sub

    '
    ' Download a remote file to a local file name which can 
    '  include a path. The local file name will be created or 
    '  overwritten, but the path must exist.
    Public Sub DownloadFile(ByVal sFileName As String, _
                            ByVal sLocalFileName As String)
        DownloadFile(sFileName, sLocalFileName, False)
    End Sub

    '
    ' Download a remote file to a local file name which can 
    '  include a path, and set the resume flag. The local file 
    '  name will be created or overwritten, but the path must 
    '  exist.
    Public Sub DownloadFile(ByVal sFileName As String, _
                            ByVal sLocalFileName As String, _
                            ByVal bResume As Boolean)
        Dim st As Stream
        Dim output As FileStream
        Dim cSocket As Socket
        Dim offset, npos As Long

        If (Not (Me.loggedIn)) Then
            Login()
        End If

        SetBinaryMode(True)

        If (sLocalFileName.Equals("")) Then
            sLocalFileName = sFileName
        End If

        If (Not (File.Exists(sLocalFileName))) Then
            st = File.Create(sLocalFileName)
            st.Close()
        End If

        output = New FileStream(sLocalFileName, FileMode.Open)
        cSocket = CreateDataSocket()
        offset = 0

        If (bResume) Then
            offset = output.Length

            If (offset > 0) Then
                SendCommand("REST " & offset)
                If (retValue <> 350) Then
                    'throw new IOException(reply.Substring(4));
                    'Some servers may not support resuming.
                    offset = 0
                End If
            End If

            If (offset > 0) Then
                npos = output.Seek(offset, SeekOrigin.Begin)
            End If
        End If

        SendCommand("RETR " & sFileName)

        If (Not (retValue = 226 Or retValue = 150 Or retValue = 125)) Then
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        Do While (True)
            Array.Clear(aBuffer, 0, aBuffer.Length)
            iBytes = cSocket.Receive(aBuffer, aBuffer.Length, 0)
            output.Write(aBuffer, 0, iBytes)

            If (iBytes <= 0) Then
                Exit Do
            End If
        Loop

        output.Close()
        If (cSocket.Connected) Then
            cSocket.Close()
        End If

    End Sub

    '
    ' Upload a file.
    Public Sub UploadFile(ByVal sFileName As String)
        Call UploadFile(sFileName, False)
    End Sub

    '
    ' Upload a file and set the resume flag.
    Public Sub UploadFile(ByVal sFileName As String, _
                          ByVal bResume As Boolean)

        Dim cSocket As Socket = Nothing

        Dim input As FileStream = Nothing
        Dim bFileNotFound As Boolean
        Dim offset As Long

        If (Not (Me.loggedIn)) Then
            Login()
        End If

        cSocket = CreateDataSocket()
        offset = 0

        If (bResume) Then
            Try
                SetBinaryMode(True)
                offset = GetFileSize(sFileName)
            Catch ex As Exception
                offset = 0
            End Try
        End If

        If (offset > 0) Then
            SendCommand("REST " & offset)
            If (retValue <> 350) Then
                'throw new IOException(reply.Substring(4));
                'Remote server may not support resuming.
                offset = 0
            End If
        End If

        SendCommand("STOR " & Path.GetFileName(sFileName))
        If (Not (retValue = 125 Or retValue = 150)) Then
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        Try
            ' Check to see if the file exists before the upload.
            bFileNotFound = False
            If (File.Exists(sFileName)) Then
                ' Open input stream to read source file
                input = New FileStream(sFileName, FileMode.Open, FileAccess.Read)
                If (offset <> 0) Then
                    input.Seek(offset, SeekOrigin.Begin)
                End If

                If Me.wasCancelled Then
                    If (cSocket.Connected) Then
                        cSocket.Close()
                    End If
                    Exit Sub
                End If

                Dim totBytes As Long = input.Length
                Dim sentBytes As Long = 0
                ' Upload the file 
                RaiseEvent FileUploadStart(Me, New ProgressableEventArgs(totBytes))

                iBytes = input.Read(aBuffer, 0, aBuffer.Length)

                Do While (iBytes > 0)
                    sentBytes += iBytes

                    Dim pevnt As ProgressableEventArgs = New ProgressableEventArgs(sentBytes, totBytes)
                    RaiseEvent FileUploadProgress(Me, pevnt)
                    If pevnt.wasCancelled Then
                        Me.wasCancelled = True
                        Exit Sub
                    End If

                    cSocket.Send(aBuffer, iBytes, 0)
                    iBytes = input.Read(aBuffer, 0, aBuffer.Length)

                Loop

                RaiseEvent FileUploadFinished(Me, New ProgressableEventArgs(totBytes))


            Else
                bFileNotFound = True
            End If

        Finally
            If (input IsNot Nothing) Then
                input.Close()
            End If

            If (cSocket.Connected) Then
                cSocket.Close()
            End If
        End Try

        ' No point in reading the return value if the file was
        '  not found.
        If (bFileNotFound) Then
            MessageString = serverReply
            Throw New IOException("The file: " & sFileName & " was not found.  Can not upload the file to the FTP Site.")
        End If

        ReadReply()
        If (Not (retValue = 226 Or retValue = 250)) Then
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

    End Sub

    '
    ' Delete a file from the remote FTP server.
    Public Function DeleteFile(ByVal sFileName As String) As Boolean
        Dim bResult As Boolean

        bResult = True
        If (Not (Me.loggedIn)) Then
            Login()
        End If

        SendCommand("DELE " & sFileName)
        If (retValue <> 250) Then
            bResult = False
            MessageString = serverReply
        End If

        ' Return the final result.
        Return bResult
    End Function

    '
    ' Rename a file on the remote FTP server.
    Public Function RenameFile(ByVal sOldFileName As String, _
                               ByVal sNewFileName As String) As Boolean
        Dim bResult As Boolean

        bResult = True
        If (Not (Me.loggedIn)) Then
            Login()
        End If

        SendCommand("RNFR " & sOldFileName)
        If (retValue <> 350) Then
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        '  known problem
        '  rnto will not take care of existing file.
        '  i.e. It will overwrite if newFileName exist
        SendCommand("RNTO " & sNewFileName)
        If (retValue <> 250) Then
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        Return bResult
    End Function

    '
    ' Create a directory on the remote FTP server.
    Public Function CreateDirectory(ByVal sDirName As String) As Boolean
        Dim bResult As Boolean

        bResult = True
        If (Not (Me.loggedIn)) Then
            Login()
        End If

        SendCommand("MKD " & sDirName)

        If (retValue <> 257) Then
            bResult = False
            MessageString = serverReply
        End If

        ' Return the final result.
        Return bResult
    End Function

    '
    ' Delete a directory on the remote FTP server.
    Public Function RemoveDirectory(ByVal sDirName As String) As Boolean
        Dim bResult As Boolean

        bResult = True
        If (Not (Me.loggedIn)) Then
            Login()
        End If

        SendCommand("RMD " & sDirName)
        If (retValue <> 250) Then
            bResult = False
            MessageString = serverReply
        End If

        ' Return the final result.
        Return bResult
    End Function

    '
    ' Change the current working directory on the remote FTP 
    '  server.
    Public Function ChangeDirectory(ByVal sDirName As String) As Boolean
        Dim bResult As Boolean

        bResult = True
        If (sDirName.Equals(".")) Then
            Return False
        End If

        If (Not (Me.loggedIn)) Then
            Login()
        End If

        SendCommand("CWD " & sDirName)
        If (retValue <> 250) Then

            If retValue = 550 Then
                Me.CreateDirectory(sDirName)
            End If

            bResult = False
            MessageString = serverReply
        End If

        Me.RemotePath = sDirName

        ' Return the final result.
        Return bResult
    End Function

    '
    ' Close the FTP connection.
    Public Sub CloseConnection()
        If (Not (objClientSocket Is Nothing)) Then
            SendCommand("QUIT")
        End If

        Cleanup()
    End Sub

#End Region

#Region "Private Subs and Functions"
    '
    ' Read the reply from the FTP Server
    Private Sub ReadReply()
        serverMessage = ""
        serverReply = ReadLine()
        retValue = Int32.Parse(serverReply.Substring(0, 3))

    End Sub

    '
    ' Clean up some variables.
    Private Sub Cleanup()
        If Not (objClientSocket Is Nothing) Then
            objClientSocket.Close()
            objClientSocket = Nothing
        End If

        Me.loggedIn = False
    End Sub

    '
    ' Read a line from the server.
    Private Function ReadLine(Optional ByVal bClearMes As Boolean = False) As String
        Dim seperator As Char = ControlChars.Lf
        Dim mess() As String

        If (bClearMes) Then
            serverMessage = ""
        End If
        Do While (True)
            Array.Clear(aBuffer, 0, BLOCK_SIZE)
            iBytes = objClientSocket.Receive(aBuffer, aBuffer.Length, 0)
            serverMessage += ASCII.GetString(aBuffer, 0, iBytes)
            If (iBytes < aBuffer.Length) Then
                Exit Do
            End If
        Loop

        mess = serverMessage.Split(seperator)
        If (serverMessage.Length > 2) Then
            serverMessage = mess(mess.Length - 2)
        Else
            serverMessage = mess(0)
        End If

        If (Not (serverMessage.Substring(3, 1).Equals(" "))) Then
            Return ReadLine(True)
        End If

        Return serverMessage
    End Function

    '
    ' Send a command to the FTP Server.
    Public Sub SendCommand(ByVal sCommand As String)

        sCommand = sCommand & ControlChars.CrLf
        Dim cmdbytes As Byte() = ASCII.GetBytes(sCommand)

        objClientSocket.Send(cmdbytes, cmdbytes.Length, 0)
        ReadReply()

        Debug.Write(sCommand)
        Debug.Write(serverReply)
        Debug.WriteLine("-----------")
    End Sub


    Private Function CreateDataSocket() As Socket

        Dim index1, index2, len As Int32
        Dim partCount, i, port As Int32
        Dim ipData As String, buf As String ', ipAddress As String
        Dim parts(6) As Int32
        Dim ch As Char
        Dim s As Socket
        Dim ep As IPEndPoint

        SendCommand("PASV")
        If (retValue <> 227) Then
            MessageString = serverReply
            Throw New IOException(serverReply.Substring(4))
        End If

        index1 = serverReply.IndexOf("(")
        index2 = serverReply.IndexOf(")")
        ipData = serverReply.Substring(index1 + 1, index2 - index1 - 1)

        len = ipData.Length
        partCount = 0
        buf = ""

        For i = 0 To (len - 1) 'And partCount <= 6)
            ch = Char.Parse(ipData.Substring(i, 1))
            If (Char.IsDigit(ch)) Then
                buf += ch
            ElseIf (ch <> ",") Then
                MessageString = serverReply
                Throw New IOException("Malformed PASV reply: " & serverReply)
            End If

            If ((ch = ",") Or (i + 1 = len)) Then
                Try
                    parts(partCount) = Int32.Parse(buf)
                    partCount += 1
                    buf = ""
                Catch ex As Exception
                    MessageString = serverReply
                    Throw New IOException("Malformed PASV reply: " & serverReply)
                End Try
            End If
        Next

        'ipAddress = parts(0) & "." & parts(1) & "." & parts(2) & "." & parts(3)

        ' Make this call in VB.NET 2002.  We would like to 
        '  bitshift the number by 8 bits, so in VB.NET 2002 we
        '  multiply the number by 2 to the power of 8.
        'port = CInt(parts(4) * (2 ^ 8))

        ' Make this call and comment out the above line for
        '  VB.NET 2003.
        port = parts(4) << 8

        ' Determine the data port number.
        port = port + parts(5)

        s = New Socket(AddressFamily.InterNetwork, SocketType.Stream, ProtocolType.Tcp)
        'ep = New IPEndPoint(Dns.Resolve(ipAddress).AddressList(0), port)
        ep = New IPEndPoint(New System.Net.IPAddress(IPAddress.Parse(Me.RemoteHost).GetAddressBytes()), port)


        Try
            s.Connect(ep)
        Catch ex As Exception
            MessageString = serverReply
            Throw New IOException("Can't connect to remote server")
        End Try

        Return s
    End Function

#End Region

#Region " *** Example Code *** "
    '
    ' Copy and paste the code below into a VB WebForm or WinForm
    '  application and then do the following:
    '
    '       1).  From within the ASP.NET or WinForm app set a
    '            reference to the FTP.dll and BitOperators.dll
    '            files.
    '       2).  At the top of the application code file 
    '            (E.g WebForm1.aspx.vb or Form1.vb) type in
    '               Imports FTP
    '       3).  Compile the application and run.
    '       4).  Have fun.

    'Protected Sub TestFTP()
    '    Dim ff As FTPClient

    '    Try
    '        '-------------------------------------------
    '        ' OPTION 1
    '        ' --------
    '        '
    '        ' Create an instance of the FTP Class.
    '        'ff = New FTPClient()

    '        ' Setup the appropriate properties.
    '        'ff.RemoteHost = "microsoft"
    '        'ff.RemoteUser = "ftpuser"
    '        'ff.RemotePassword = "password"
    '        '-------------------------------------------

    '        '-------------------------------------------
    '        ' OPTION 2
    '        ' --------
    '        '  Pass the values into the constructor 
    '        '  instead.  These can be overridden by simply
    '        '  setting the appropriate properties on the
    '        '  instance of the FTPClient Class.
    '        ff = New FTPClient("microsoft", _
    '                        ".", _
    '                        "ftpuser", _
    '                        "password", _
    '                        21)

    '        ' Attempt to log into the FTP Server.
    '        If (ff.Login()) Then
    '            '
    '            ' Move the to Area1\Section1\Subby1\ directory.
    '            ff.ChangeDirectory("Area1")
    '            ff.ChangeDirectory("Section1")

    '            'ff.CreateDirectory("Subby1")
    '            ff.ChangeDirectory("Subby1")
    '            ff.SetBinaryMode(True)

    '            ' Upload a file.
    '            'ff.UploadFile("d:\general\secureapps.pdf")

    '            ' Download a file.
    '            'ff.DownloadFile("secureapps.pdf", "d:\general\secureapps.pdf")

    '            ' Remove a file from the FTP Site.
    '            If (ff.DeleteFile("secureapps.pdf")) Then
    '                Response.Write("File has been removed from FTP Site")
    '                'MessageBox.Show("File has been removed from FTP Site")
    '            Else
    '                Response.Write("Unable to remove file from FTP Site.  Message from server: " & ff.MessageString & "<br>")
    '                'MessageBox.Show("Unable to remove file from FTP Site")
    '            End If

    '            ' Rename a file on the FTP Site.
    '            'If (ff.RenameFile("secureapps.pdf", "newapp.pdf")) Then
    '            '    Response.Write("File has been renamed")
    '            '    MessageBox.Show("File has been renamed")
    '            'End If

    '            'ff.ChangeDirectory("..")
    '            'If (ff.RemoveDirectory("Subby1")) Then
    '            '    Response.Write("Directory has been removed<br>")
    '            '    ' MessageBox.Show("Directory has been removed")
    '            'Else
    '            '    Response.Write("Unable to remove the directory.  Message from server: " & ff.MessageString & "<br>")
    '            '    ' MessageBox.Show("Unable to remove the directory.")
    '            'End If
    '        End If

    '    Catch ex As System.Exception
    '        ' ASP.NET
    '        Response.Write(ex.Message & "<BR>")
    '        Response.Write("Message from FTP Server was: " & ff.MessageString)

    '        ' WinForms
    '        'Messagebox.Show(ex.Message)
    '        'MessageBox.show("Message from FTP Server was: " & ff.MessageString)
    '    Finally
    '        '
    '        ' Always close down the connection to ensure that
    '        '  there are no "stray" Fido's Fetching data.  In
    '        '  other words, no stray/limbo/not-in-use FTP
    '        '  connections.
    '        ff.CloseConnection()
    '    End Try
    'End Sub

#End Region

    Public Sub startProcess() Implements IProgressable.startProcess

    End Sub
End Class


